
      **********************************************************        00010002 

      **  SAMPLE CALLABLE SERVICE PROGRAM.                    **        00020002 

      **  SHARE SESSION: INTRODUCING LE CALLABLE SERVICES     **        00030002 

      **  THOMAS PETROLINO                                    **        00040002 

      **  IBM LANGUAGE ENVIRONMENT                            **        00050002 

      **  TAPETRO@US.IBM.COM                                  **        00051002 

      **********************************************************        00052002 

       IDENTIFICATION DIVISION.                                         00060000 

       PROGRAM-ID. CSSQRTCB.                                            00070004 

                                                                        00080000 

       DATA DIVISION.                                                   00090000 

       WORKING-STORAGE SECTION.                                         00100000 

      **********************************************************        00110002 

      **  DEFINE SPACE FOR A LANGUAGE ENVIRONMENT FEEDBACK    **        00120002 

      **  CODE. 12 TOTAL BYTES                                **        00120102 

      **  INCLUDE A COPY OF CEEIGZCT TO GET PRE-DEFINED       **        00120202 

      **  LANGUAGE ENVIRONMENT CONDITION TOKENS.              **        00120302 

      **********************************************************        00120402 

        01  FC.                                                         00120502 

            02  CONDITION-TOKEN-VALUE.                                  00120602 

            COPY  CEEIGZCT.                                             00120702 

                03  CASE-1-CONDITION-ID.                                00120802 

                   04  SEVERITY     PIC S9(4) BINARY.                   00120902 

                   04  MSG-NO       PIC S9(4) BINARY.                   00121002 

                03  CASE-SEV-CTL    PIC X.                              00121100 

                03  FACILITY-ID     PIC XXX.                            00121200 

            02  I-S-INFO            PIC S9(9) BINARY.                   00121300 

      **********************************************************        00121402 

      **  DEFINE SOME STORAGE TO BE USED BY THE LANGUAGE      **        00121502 

      **  ENVIRONMENT MATH CALLABLE SERVICES.                 **        00121602 

      **********************************************************        00121902 

       01 MY-INPUT                  COMP-1.                             00126000 

       01 RESULT                    COMP-1.                             00129000 

       01 RESULT2                   PIC ZZZZ.99.                        00129100 

      **********************************************************        00129202 

      **  DEFINE SOME STORAGE TO BE USED BY THE LANGUAGE      **        00129302 

      **  ENVIRONMENT MESSAGING CALLABLE SERVICES             **        00129402 

      **********************************************************        00129502 

       01 MSG-DEST                  PIC S9(9) BINARY VALUE 2.           00129600 

       01 MY-MSG-STRING.                                                00129700 

          02  MY-STRING-LEN         PIC S9(4) BINARY.                   00129800 

          02  MY-STRING-TEXT.                                           00129900 

             03  MY-STRING-CHAR     PIC X(256).                         00130005 

       01 MY-OUTPUT-STRING.                                             00130403 

         02 MY-OUTPUT-STRING-LEN    PIC S9(4) BINARY VALUE 38.          00130503 

         02 MY-OUTPUT-STRING-TEXT   PIC X(19)                           00130603 

                                      VALUE "THE SQUARE ROOT OF ".      00130703 

         02 MY-OUTPUT-STRING-INPUT  PIC ZZZ9.99.                        00130803 

         02 MY-OUTPUT-STRING-TEXT2  PIC X(5) VALUE " IS: ".             00130903 

         02 MY-OUTPUT-STRING-RESULT PIC ZZZ9.99.                        00131003 

      **********************************************************        00131102 

      **  DEFINE SOME STORAGE TO BE USED BY THE LANGUAGE      **        00131202 

      **  ENVIRONMENT CEE3DMP CALLABLE SERVICE                **        00131302 

      **********************************************************        00131402 

       01 DUMP-TITLE                PIC X(80).                          00131500 

       01 DUMP-OPTIONS              PIC X(255).                         00131600 

      **********************************************************        00131702 



      **  DEFINE SOME STORAGE TO BE USED BY THE LANGUAGE      **        00131802 

      **  ENVIRONMENT DATE/TIME CALLABLE SERVICES             **        00131902 

      **********************************************************        00132002 

       01 MY-DATE-LILIAN            PIC S9(9) BINARY.                   00132101 

       01 MY-SECS-LILIAN            COMP-2.                             00132201 

       01 MY-TIME-GREGORIAN         PIC X(17).                          00132301 

       01 MY-PIC-STRING.                                                00132401 

          02  MY-PIC-LEN            PIC S9(4) BINARY.                   00132505 

          02  MY-PIC-TEXT.                                              00132605 

             03  MY-PIC-CHAR        PIC X(256).                         00132705 

       01 MY-TIMESTAMP              PIC X(80).                          00133101 

      **********************************************************        00133206 

      **  DEFINE SOME STORAGE TO BE USED BY CEE3ABD           **        00133306 

      **********************************************************        00133506 

       01  ABDCODE                 PIC S9(9) BINARY.                    00133606 

       01  TIMING                  PIC S9(9) BINARY.                    00133706 

      **********************************************************        00133802 

      **  START SOME REAL CODE NOW...                         **        00133902 

      **********************************************************        00134002 

       PROCEDURE DIVISION.                                              00135000 

       MAIN-PROG.                                                       00140000 

      *        *************************************************        00140102 

      *        ** SET UP DUMP-TITLE AND DUMP-OPTIONS FOR      **        00140202 

      *        ** CEE3DMP CALL.                               **        00140302 

      *        *************************************************        00140402 

                MOVE "SAMPLE DUMP TAKEN BY CEE3DMP." TO DUMP-TITLE.     00140502 

                MOVE "NOCOND" TO DUMP-OPTIONS.                          00140602 

      *        *************************************************        00140706 

      *        ** SET UP ABDCODE AND TIMING FOR CEE3ABD CALL  **        00140806 

      *        *************************************************        00141006 

                MOVE 1234 TO ABDCODE.                                   00141106 

                MOVE 0 TO TIMING.                                       00141206 

      *        *************************************************        00141302 

      *        ** NOW CALL CEELOCT. IT RETURNS LOCAL TIME AS: **        00141402 

      *        **    DAYS SINCE OCT 14, 1582.                 **        00141502 

      *        **    SECONDS SINCE OCT 14, 1582               **        00141602 

      *        **    TIME STRING OF THE FORM YYYYMMDDHHMISS999**        00141702 

      *        **    FEEDBACK CODE                            **        00141802 

      *        ** REPORT ERROR AND STOP IF CEELOCT FAILS.     **        00141902 

      *        *************************************************        00142002 

                CALL "CEELOCT" USING MY-DATE-LILIAN, MY-SECS-LILIAN,    00142102 

                                     MY-TIME-GREGORIAN, FC.             00142202 

                IF NOT CEE000 OF FC THEN                                00142302 

      *           **********************************************        00142402 

      *           ** USE CEEMSG TO DISPLAY LANG. ENV. MESSAGE **        00142502 

      *           **********************************************        00142602 

                   CALL "CEEMSG" USING FC, MSG-DEST, OMITTED            00142702 

      *           **********************************************        00142802 

      *           ** CALL CEE3DMP TO PRODUCE CEEDUMP          **        00142902 

      *           **********************************************        00143002 

                   CALL "CEE3DMP" USING DUMP-TITLE, DUMP-OPTIONS, FC    00143102 

                   CALL "CEE3ABD" USING ABDCODE , TIMING.               00143206 

                END-IF.                                                 00143302 

      *        *************************************************        00143402 

      *        ** MOVE A PICTURE STRING INTO MY-PIC-STRING SO **        00143502 

      *        ** WE CAN CALL CEEDATM TO CONVERT THE TIME     **        00143602 

      *        *************************************************        00143702 



                MOVE 22 TO MY-PIC-LEN.                                  00143805 

                MOVE "MM/DD/YYYY HH:MI:SS AP" TO MY-PIC-TEXT.           00143905 

      *        *************************************************        00144002 

      *        ** NOW CALL CEEDATM. INPUT: LILIAN SECONDS     **        00144102 

      *        **                          PICTURE STRING     **        00144202 

      *        **                   IT RETURNS:               **        00144302 

      *        **    INPUT DATE/TIME AS A CHARACTER STRING.   **        00144402 

      *        **    FEEDBACK CODE                            **        00144502 

      *        ** REPORT ERROR AND STOP IF CEEDATM FAILS.     **        00144602 

      *        *************************************************        00144702 

                CALL "CEEDATM" USING MY-SECS-LILIAN,                    00144802 

                                     MY-PIC-STRING,                     00144902 

                                     MY-TIMESTAMP, FC.                  00145002 

                IF NOT CEE000 OF FC THEN                                00145102 

      *           **********************************************        00145202 

      *           ** USE CEEMSG TO DISPLAY LANG. ENV. MESSAGE **        00145302 

      *           **********************************************        00145402 

                   CALL "CEEMSG" USING FC, MSG-DEST, OMITTED            00145502 

      *           **********************************************        00145602 

      *           ** CALL CEE3DMP TO PRODUCE CEEDUMP          **        00145702 

      *           **********************************************        00145802 

                   CALL "CEE3DMP" USING DUMP-TITLE, DUMP-OPTIONS, FC    00145902 

                   CALL "CEE3ABD" USING ABDCODE , TIMING                00146006 

                END-IF.                                                 00146102 

      *        *************************************************        00146202 

      *        ** MOVE THE RESULT FROM CEEDATM INTO A VARYING **        00146302 

      *        ** STRING SO WE CAN USE CEEMOUT TO OUTPUT IT.  **        00146402 

      *        *************************************************        00146502 

                MOVE 22 TO MY-STRING-LEN.                               00146605 

                MOVE MY-TIMESTAMP TO MY-STRING-TEXT.                    00146705 

      *        *************************************************        00146802 

      *        ** NOW CALL CEEMOUT. INPUT: STRING TO WRITE    **        00146902 

      *        **                          DESTINATION        **        00147002 

      *        **                   IT RETURNS:               **        00147102 

      *        **    FEEDBACK CODE                            **        00147202 

      *        ** REPORT ERROR AND STOP IF CEEMOUT FAILS.     **        00147302 

      *        *************************************************        00147402 

                CALL "CEEMOUT" USING MY-MSG-STRING, MSG-DEST, FC.       00147502 

                IF NOT CEE000 OF FC THEN                                00147602 

      *           **********************************************        00147702 

      *           ** USE CEEMSG TO DISPLAY LANG. ENV. MESSAGE **        00147802 

      *           **********************************************        00147902 

                   CALL "CEEMSG" USING FC, MSG-DEST, OMITTED            00148002 

      *           **********************************************        00148102 

      *           ** CALL CEE3DMP TO PRODUCE CEEDUMP          **        00148202 

      *           **********************************************        00148302 

                   CALL "CEE3DMP" USING DUMP-TITLE, DUMP-OPTIONS, FC    00148402 

                   CALL "CEE3ABD" USING ABDCODE , TIMING                00148506 

                END-IF.                                                 00148602 

      *        *************************************************        00148702 

      *        ** MOVE STARTUP STRING INTO OUTPUT STRING USED **        00148802 

      *        ** PREVIOUSLY SO WE CAN OUTPUT IT WITH CEEMOUT **        00148902 

      *        *************************************************        00149002 

                MOVE 16 TO MY-STRING-LEN.                               00149105 

                MOVE "IN MAIN PROGRAM!" TO MY-STRING-TEXT.              00150005 

      *        *************************************************        00152002 

      *        ** NOW CALL CEEMOUT. INPUT: STRING TO WRITE    **        00153002 



      *        **                          DESTINATION        **        00154002 

      *        **                   IT RETURNS:               **        00155002 

      *        **    FEEDBACK CODE                            **        00156002 

      *        ** REPORT ERROR AND STOP IF CEEMOUT FAILS.     **        00157002 

      *        *************************************************        00158002 

                CALL "CEEMOUT" USING MY-MSG-STRING, MSG-DEST, FC.       00170000 

                IF NOT CEE000 OF FC THEN                                00180000 

      *           **********************************************        00181002 

      *           ** USE CEEMSG TO DISPLAY LANG. ENV. MESSAGE **        00182002 

      *           **********************************************        00183002 

                   CALL "CEEMSG" USING FC, MSG-DEST, OMITTED            00184002 

      *           **********************************************        00185002 

      *           ** CALL CEE3DMP TO PRODUCE CEEDUMP          **        00186002 

      *           **********************************************        00187002 

                   CALL "CEE3DMP" USING DUMP-TITLE, DUMP-OPTIONS, FC    00191000 

                   CALL "CEE3ABD" USING ABDCODE , TIMING                00200006 

                END-IF.                                                 00210000 

      *        *************************************************        00210102 

      *        ** SETUP TO CALL CEESSSQT WITH 9 AS INPUT.     **        00210202 

      *        *************************************************        00210402 

                MOVE 9 TO MY-INPUT.                                     00210500 

      *        *************************************************        00210602 

      *        ** NOW CALL CEESSSQT. INPUT: 9                 **        00210702 

      *        **                   IT RETURNS:               **        00210902 

      *        **    FEEDBACK CODE                            **        00211002 

      *        **    RESULT                                   **        00211102 

      *        ** REPORT ERROR AND STOP IF CEESSSQT FAILS.    **        00211202 

      *        *************************************************        00211302 

                CALL "CEESSSQT" USING MY-INPUT, FC, RESULT.             00211402 

                IF NOT CEE000 OF FC THEN                                00211502 

      *           **********************************************        00211602 

      *           ** USE CEEMSG TO DISPLAY LANG. ENV. MESSAGE **        00211702 

      *           **********************************************        00211802 

                   CALL "CEEMSG" USING FC, MSG-DEST, OMITTED            00211902 

      *           **********************************************        00212002 

      *           ** CALL CEE3DMP TO PRODUCE CEEDUMP          **        00212102 

      *           **********************************************        00212202 

                   CALL "CEE3DMP" USING DUMP-TITLE, DUMP-OPTIONS, FC    00212402 

                   CALL "CEE3ABD" USING ABDCODE , TIMING                00212506 

                END-IF.                                                 00212602 

      *        *************************************************        00212702 

      *        ** DISPLAY THE OUTPUT.                         **        00212802 

      *        *************************************************        00212902 

                MOVE RESULT TO MY-OUTPUT-STRING-RESULT.                 00213003 

                MOVE MY-INPUT TO MY-OUTPUT-STRING-INPUT.                00213103 

      *        *************************************************        00213203 

      *        ** NOW CALL CEEMOUT. INPUT: STRING TO WRITE    **        00213303 

      *        **                          DESTINATION        **        00213403 

      *        **                   IT RETURNS:               **        00213503 

      *        **    FEEDBACK CODE                            **        00213603 

      *        ** REPORT ERROR AND STOP IF CEEMOUT FAILS.     **        00213703 

      *        *************************************************        00213803 

                CALL "CEEMOUT" USING MY-OUTPUT-STRING, MSG-DEST, FC.    00213903 

                IF NOT CEE000 OF FC THEN                                00214003 

      *           **********************************************        00214103 

      *           ** USE CEEMSG TO DISPLAY LANG. ENV. MESSAGE **        00214203 

      *           **********************************************        00214303 



                   CALL "CEEMSG" USING FC, MSG-DEST, OMITTED            00214403 

      *           **********************************************        00214503 

      *           ** CALL CEE3DMP TO PRODUCE CEEDUMP          **        00214603 

      *           **********************************************        00214703 

                   CALL "CEE3DMP" USING DUMP-TITLE, DUMP-OPTIONS, FC    00214803 

                   CALL "CEE3ABD" USING ABDCODE , TIMING                00214906 

                END-IF.                                                 00215003 

      *        *************************************************        00215203 

      *        ** SETUP TO CALL CEESSSQT WITH 144 AS INPUT.   **        00215303 

      *        *************************************************        00215403 

                MOVE 144 TO MY-INPUT.                                   00215503 

      *        *************************************************        00215603 

      *        ** NOW CALL CEESSSQT. INPUT: 144               **        00215703 

      *        **                   IT RETURNS:               **        00215803 

      *        **    FEEDBACK CODE                            **        00215903 

      *        **    RESULT                                   **        00216003 

      *        ** REPORT ERROR AND STOP IF CEESSSQT FAILS.    **        00216103 

      *        *************************************************        00216203 

                CALL "CEESSSQT" USING MY-INPUT, FC, RESULT.             00216303 

                IF NOT CEE000 OF FC THEN                                00216403 

      *           **********************************************        00216503 

      *           ** USE CEEMSG TO DISPLAY LANG. ENV. MESSAGE **        00216603 

      *           **********************************************        00216703 

                   CALL "CEEMSG" USING FC, MSG-DEST, OMITTED            00216803 

      *           **********************************************        00216903 

      *           ** CALL CEE3DMP TO PRODUCE CEEDUMP          **        00217003 

      *           **********************************************        00217103 

                   CALL "CEE3DMP" USING DUMP-TITLE, DUMP-OPTIONS, FC    00217202 

                   CALL "CEE3ABD" USING ABDCODE , TIMING                00217306 

                END-IF.                                                 00217402 

      *        *************************************************        00217502 

      *        ** DISPLAY THE OUTPUT.                         **        00217602 

      *        *************************************************        00217702 

                MOVE RESULT TO MY-OUTPUT-STRING-RESULT.                 00217803 

                MOVE MY-INPUT TO MY-OUTPUT-STRING-INPUT.                00217903 

      *        *************************************************        00218003 

      *        ** NOW CALL CEEMOUT. INPUT: STRING TO WRITE    **        00218103 

      *        **                          DESTINATION        **        00218203 

      *        **                   IT RETURNS:               **        00218303 

      *        **    FEEDBACK CODE                            **        00218403 

      *        ** REPORT ERROR AND STOP IF CEEMOUT FAILS.     **        00218503 

      *        *************************************************        00218603 

                CALL "CEEMOUT" USING MY-OUTPUT-STRING, MSG-DEST, FC.    00218703 

                IF NOT CEE000 OF FC THEN                                00218803 

      *           **********************************************        00218903 

      *           ** USE CEEMSG TO DISPLAY LANG. ENV. MESSAGE **        00219003 

      *           **********************************************        00219103 

                   CALL "CEEMSG" USING FC, MSG-DEST, OMITTED            00219203 

      *           **********************************************        00219303 

      *           ** CALL CEE3DMP TO PRODUCE CEEDUMP          **        00219403 

      *           **********************************************        00219503 

                   CALL "CEE3DMP" USING DUMP-TITLE, DUMP-OPTIONS, FC    00219603 

                   CALL "CEE3ABD" USING ABDCODE , TIMING                00219706 

                END-IF.                                                 00219803 

      *        *************************************************        00220102 

      *        ** SETUP TO CALL CEESSSQT WITH 2500 AS INPUT.  **        00220202 

      *        *************************************************        00220302 



                MOVE 2500 TO MY-INPUT                                   00220402 

      *        *************************************************        00220502 

      *        ** NOW CALL CEESSSQT. INPUT: 2500              **        00220602 

      *        **                   IT RETURNS:               **        00220702 

      *        **    FEEDBACK CODE                            **        00220802 

      *        **    RESULT                                   **        00220902 

      *        ** REPORT ERROR AND STOP IF CEESSSQT FAILS.    **        00221002 

      *        *************************************************        00221102 

                CALL "CEESSSQT" USING MY-INPUT, FC, RESULT.             00221202 

                IF NOT CEE000 OF FC THEN                                00221302 

                   CALL "CEEMSG" USING FC, MSG-DEST, OMITTED            00221402 

                   CALL "CEE3DMP" USING DUMP-TITLE, DUMP-OPTIONS, FC    00221502 

                   CALL "CEE3ABD" USING ABDCODE , TIMING                00221606 

                END-IF.                                                 00221702 

      *        *************************************************        00221802 

      *        ** DISPLAY THE OUTPUT.                         **        00221902 

      *        *************************************************        00222002 

                MOVE RESULT TO MY-OUTPUT-STRING-RESULT.                 00222103 

                MOVE MY-INPUT TO MY-OUTPUT-STRING-INPUT.                00222203 

      *        *************************************************        00222303 

      *        ** NOW CALL CEEMOUT. INPUT: STRING TO WRITE    **        00222403 

      *        **                          DESTINATION        **        00222503 

      *        **                   IT RETURNS:               **        00222603 

      *        **    FEEDBACK CODE                            **        00222703 

      *        ** REPORT ERROR AND STOP IF CEEMOUT FAILS.     **        00222803 

      *        *************************************************        00222903 

                CALL "CEEMOUT" USING MY-OUTPUT-STRING, MSG-DEST, FC.    00223003 

                IF NOT CEE000 OF FC THEN                                00223103 

      *           **********************************************        00223203 

      *           ** USE CEEMSG TO DISPLAY LANG. ENV. MESSAGE **        00223303 

      *           **********************************************        00223403 

                   CALL "CEEMSG" USING FC, MSG-DEST, OMITTED            00223503 

      *           **********************************************        00223603 

      *           ** CALL CEE3DMP TO PRODUCE CEEDUMP          **        00223703 

      *           **********************************************        00223803 

                   CALL "CEE3DMP" USING DUMP-TITLE, DUMP-OPTIONS, FC    00223903 

                   CALL "CEE3ABD" USING ABDCODE , TIMING                00224006 

                END-IF.                                                 00224103 

      *        *************************************************        00224402 

      *        ** SETUP TO CALL CEESSSQT WITH -99 AS INPUT.   **        00224502 

      *        **   WE SHOULD EXPECT TO SEE AN ERROR HERE.    **        00224602 

      *        *************************************************        00224702 

                MOVE -99 TO MY-INPUT.                                   00224802 

      *        *************************************************        00224902 

      *        ** NOW CALL CEESSSQT. INPUT: -99               **        00225002 

      *        **                   IT RETURNS:               **        00225102 

      *        **    FEEDBACK CODE                            **        00225202 

      *        **    RESULT                                   **        00225302 

      *        ** REPORT ERROR AND STOP IF CEESSSQT FAILS.    **        00225402 

      *        *************************************************        00225502 

                CALL "CEESSSQT" USING MY-INPUT, FC, RESULT.             00225602 

                IF NOT CEE000 OF FC THEN                                00225702 

                   CALL "CEEMSG" USING FC, MSG-DEST, OMITTED            00225802 

                   CALL "CEE3DMP" USING DUMP-TITLE, DUMP-OPTIONS, FC    00225902 

                   CALL "CEE3ABD" USING ABDCODE , TIMING                00226006 

                END-IF.                                                 00226102 

      *        *************************************************        00226202 



      *        ** DISPLAY THE OUTPUT.                         **        00226302 

      *        *************************************************        00226402 

                MOVE RESULT TO MY-OUTPUT-STRING-RESULT.                 00226503 

                MOVE MY-INPUT TO MY-OUTPUT-STRING-INPUT.                00226603 

      *        *************************************************        00226703 

      *        ** NOW CALL CEEMOUT. INPUT: STRING TO WRITE    **        00226803 

      *        **                          DESTINATION        **        00226903 

      *        **                   IT RETURNS:               **        00227003 

      *        **    FEEDBACK CODE                            **        00227103 

      *        ** REPORT ERROR AND STOP IF CEEMOUT FAILS.     **        00227203 

      *        *************************************************        00227303 

                CALL "CEEMOUT" USING MY-OUTPUT-STRING, MSG-DEST, FC.    00227403 

                IF NOT CEE000 OF FC THEN                                00227503 

      *           **********************************************        00227603 

      *           ** USE CEEMSG TO DISPLAY LANG. ENV. MESSAGE **        00227703 

      *           **********************************************        00227803 

                   CALL "CEEMSG" USING FC, MSG-DEST, OMITTED            00227903 

      *           **********************************************        00228003 

      *           ** CALL CEE3DMP TO PRODUCE CEEDUMP          **        00228103 

      *           **********************************************        00228203 

                   CALL "CEE3DMP" USING DUMP-TITLE, DUMP-OPTIONS, FC    00228303 

                   CALL "CEE3ABD" USING ABDCODE , TIMING                00228406 

                END-IF.                                                 00228503 

      *        *************************************************        00228802 

      *        ** WE ARE OUTTA HERE!                          **        00228902 

      *        *************************************************        00229002 

                GOBACK.                                                 00240000 � 

